Tips and Tricks for Creating the Reports Your Clients Need to See

Agenda

- Produce a “typical” Demographics table
  - Produce output according to a pre-defined layo
  - “JAZZ IT UP” with ODS styles
- Goals
  - Add to your bag of tricks
  - Bridge the gap between PROC REPORT 101 and real-world applications
  - Discover how to quickly get answers from documentation
- PROC REPORT tools
  - ACROSS variables
  - CALL DEFINE
  - Temporary variables
- ODS tools
  - PROC REPORT-specific styling
  - In-line formatting
  - Using destination-specific language

DEMOGRAPHICS

	Treatment Group			
	Placebo (N=xxx)	Treatment A (N=xxx)	Treatment B (N=xxx)	p-value
Age				
n	xxx	xxx	xxx	
Mean (SD)	xxx	xxx	xxx	xxxx
Median	xxx	xxx	xxx	
Min – Max	xxx	xxx	xxx	
Gender				
Male	xxx	xxx	xxx	xxxx
Female	xxx	xxx	xxx	
Race				
Caucasian	xxx	xxx	xxx	xxxx
Black	xxx	xxx	xxx	
Others	xxx	xxx	xxx	
Asian	xxx	xxx	xxx	
Native Hawaiian or other Pacific Islander	xxx	xxx	xxx	

Post text (e.g. explanation of p-values)

ADVERSE EVENTS

System Organ Class High Level Term	Placebo (N=xx) Subject n (%)	 Event n	Treatment A (N=xx) Subject n (%)	 Event n	Treatment B (N=xx) Subject n (%)	 Event n
System Organ Class #1	x (xx.x%)	x	x (xx.x%)	x	x (xx.x%)	x
High Level Term #1	x (xx.x%)	x	x (xx.x%)	x	x (xx.x%)	x
High Level Term #2	x (xx.x%)	x	x (xx.x%)	x	x (xx.x%)	x
High Level Term #3	x (xx.x%)	x	x (xx.x%)	x	x (xx.x%)	x
…						
System Organ Class #2	x (xx.x%)	x	x (xx.x%)	x	x (xx.x%)	x
High Level Term #4	x (xx.x%)	x	x (xx.x%)	x	x (xx.x%)	x
High Level Term #5	x (xx.x%)	x	x (xx.x%)	x	x (xx.x%)	x
High Level Term #6	x (xx.x%)	x	x (xx.x%)	x	x (xx.x%)	x

Typical DEMOGRAPHICS data set

Subject	treatment	Gender	Race	Age
001	Placebo	M	Caucasian	17
002	Placebo	F	Black	15
003	Treatment B	F	Asian	16
004	Placebo	M	Black	15
005	Placebo	M	Caucasian	14
006	Treatment A	M	Caucasian	15
007	Treatment A	M	Caucasian	17
008	Treatment B	F	Black	14
	….	...	…	...

DEMOGRAPHICS




```
ods rtf file='demo1-across1.rtf' style=minimal ;
proc report nowindows data=temp.demo ;
column order1 order2 text … ;

define order1 / group noprint order=internal ;
define order2 / group noprint order=internal ;
define text / group ' ' ;
…

compute before order1 ;
  line @1 ' ' ;
  line @1 order1 head. ;
endcomp;

run;
ods rtf close;
```

ACROSS Variables

- Each value of the variable becomes a column in the table
- Formatted values become the column headers and the columns are ordered from left to right by these formatted values
- All column are treated the same
- Most commonly used by “stacking” or interacting with other variables or statistics.
- Comparisons with other procedure variables
    - PROC TABULATE - Class variables in the column dimension
    - PROC TRANSPOSE – ID variable

See documentation

TEMP.DEMO

ACROSS Variables

treatment	order1	order2	text	values
Placebo	2	1	Male	5
Placebo	2	2	Female	2
Treatment A	1	4	Min - Max	11 - 16
Treatment B	3	3	Others	2
Placebo	3	2	Black	1
Treatment B	1	1	n	6
Treatment A	2	2	Female	3
Treatment A	3	1	Caucasian	2
Treatment B	1	2	Mean (SD)	13.2 (1.33)
….	...	…	…	…

```
proc transpose data=temp.demo out=tran;
by order1 order2 text ;
var values ;
id treatment ;
run;
```

```
ods rtf file='demo1-across1.rtf' style=minimal ;
proc report nowindows data=temp.demo ;
column order1 order2 text  treatment , values ;
define order1 / group noprint order=internal ;
define order2 / group noprint order=internal ;
define text / group ' ' ;
define treatment / across 'Treatment Group'
  order = internal format = treat. ;
define values / group ' ' ;
compute before order1 ;
  line @1 ' ' ;
  line @1 order1 head. ;
endcomp;
run;
ods rtf close;
```

TEMP.CLASS2

ACROSS Variables

treatment	order1	order2	text	values	p value
Placebo	2	1	Male	5	.4794
Placebo	2	2	Female	2	
Treatment A	1	4	Min - Max	11 - 16	
Treatment B	3	3	Others	2	
Placebo	3	2	Black	1	
Treatment B	1	1	n	6	
Treatment A	2	2	Female	3	
Treatment A	3	1	Caucasian	2	
Treatment B	1	2	Mean (SD)	13.2 (	
….	...	…	…	..	

```
ods rtf file='demo1-across1.rtf' style=minimal ;
proc report nowindows data=temp.class2 ;
column order1 order2 text treatment,values pvalue ;
define order1 / group noprint order=internal ;
define order2 / group noprint order=internal ;
define text / group ' ' ;
define treatment / across 'Treatment Group'
      order = internal format = treat. ;
define values / group ' ' ;
define pvalue / sum 'p value' format=pval. ;

compute before order1 ;
 line @1 ' ' ;
 line @1 order1 head. ;
endcomp;
run;
ods rtf close;
```

See demo1-across1.rtf

Before

```
column order1 order2 text treatment,values pvalue
…
define pvalue / sum ‘p value’ format=pval. ;
…
```

After

```
ods rtf file='demo2-across2.rtf' style=minimal ;
proc report nowindows data=temp.class2 ;
column order1 order2 text treatment,values (‘p value’ pvalue) ;
define order1 / group noprint order=internal ;
define order2 / group noprint order=internal ;
define text / group '' ;
define treatment / across ‘Treatment Group’
  order = internal format = treat.  ;
define values / group ‘ ‘ ;
define pvalue / sum ‘ ’ format=pval.  ;

compute before order1 ;
  line @1 ‘ ‘ ;
  line @1 order1 head. ;
endcomp;
run;
ods rtf close;
```

See demo2-across2.rtf

ADVERSE EVENTS

System Organ Class High Level Term	Placebo (N=xx) Subject n (%)	 Event n	Treatment A (N=xx) Subject n (%)	 Event n	Treatment B (N=xx) Subject n (%)	 Event n
System Organ Class #1	x (xx.x%)	x	x (xx.x%)	x	x (xx.x%)	x
High Level Term #1	x (xx.x%)	x	x (xx.x%)	x	x (xx.x%)	x

```
proc transpose data=temp.demo out=sub;
by order1 order2 text ;
var subjects ;
id treatment ;
run;
```

```
proc transpose data=temp.demo out=ev;
by order1 order2 text ;
var events ;
id treatment ;
run;
```

```
/* without headers */    columns … treatmnt , (subjects events) … ;
```

Now merge SUB with EV by ORDER1, ORDER, TEXT

```
/* with headers */   columns … treatmnt , ((“Subject~n (%)” subjects) (“Event~n” events) ) … ;
```

"JAZZ IT UP" IN STYLE!!

- What's in a style?
  - Formatting
  - Non-data text
- PROC REPORT styling
  - Syntax
  - location, location, location
  - Which statements allow for styling?
- Other alternatives
  - Destination language
  - In-line formatting
- Conditional styling
  - CALL DEFINE
  - Temporary variables

See documentation

DEMOGRAPHICS

"JAZZ IT UP" IN STYLE!!

	Treatment Group			
	Placebo (N=xxx)	Treatment A (N=xxx)	Treatment B (N=xxx)	p-value
Age				
n	xxx	xxx	xxx	
Mean (SD)	xxx	xxx	xxx	xxxx
Median	xxx	xxx	xxx	
Min – Max	xxx	xxx	xxx	
Gender				
Male	xxx	xxx	xxx	xxxx
Female	xxx	xxx	xxx	
Race				
Caucasian	xxx	xxx	xxx	xxxx
Black	xxx	xxx	xxx	
Others	xxx	xxx	xxx	
Asian	xxx	xxx	xxx	
Native Hawaiian or other Pacific Islander	xxx	xxx	xxx	

Post text (e.g. explanation of p-values)

Current state of the report

"JAZZ IT UP" IN STYLE!!

	Treatment Group			
	Placebo (N=7)	Treatment A (N=6)	Treatment B (N=6)	p-value
Age	Center	Center	Center	Center
N	3	2	2	
Mean (SD)	0	1	0	0.7285
Median	1	2	2	
Min - Max	1	0	0	
Gender				
Male	5	3	2	0.4794
Female	2	3	4	
Race				
Caucasian	5	0	1	0.6589
Black	1	0	0	
Others	0	2	1	
Asian	1	3	2	
Native Hawaiian or other Pacific Islander	0	1	2	

Bold

Indent

Indent

Indent

Indent

Bold

Indent

Indent

Bold

Indent

Indent

Indent

Indent

Indent

Cell width

Other stuff: All those borders; font; post-text

See documentation

```
proc report nowindows data=temp.demo
    style = [
            frame=hsides
            rules=groups
            posttext="Explanation of p values"
            font_face="times new roman"
            font_size=8 pt]
    style(header lines) = [
            font_face="times new roman"]
    style(column) = [
            font_face="times new roman"
             cellwidth=1.25 in
             just=center]  ;
```

```
column order1 order2 text treatment,counts (‘p value’ pvalue) ;
define order1 / group noprint order=internal ;
define order2 / group noprint order=internal ;
define text / group ‘ ‘ style = [cellwidth=2 in just=left indent=.25 in]  ;
define treatment / across ‘Treatment Group’  order = internal ;
define counts / group ‘ ‘ ;
define pvalue / sum ‘ ’ format=pval.  ;
```

```
compute before order1  / style = [font_weight=bold]  ;
  line @1 ‘ ‘ ;
  line @1 order1 head. ;
endcomp;
```

```
proc report nowindows data=temp.demo
   style = [frame=hsides rules=groups posttext="Explanation of p values"
            font_face="times new roman" font_size=8 pt just=left]
   style(header lines) = [font_face="times new roman"]
   style(column) = [font_face="times new roman" cellwidth=1.25 in just=center] ;
column order1 order2 text treatment,counts ('p value' pvalue) ;
define order1 / group noprint order=internal ;
define order2 / group noprint order=internal ;
define text / group ' '   style = [cellwidth=3 in just=left indent=.25 in] ;
define treatment / across 'Treatment Group'
  order = internal ;
define counts / group ' ' ;
define pvalue / sum ' ' format=pval. ;
compute before order1  / style = [font_weight=bold] ;
  line @1 ' ' ;
  line @1 order1 head. ;
endcomp;
```

See demo3-rptstyle1.rtf

See demo4-rptstyle2.rtf

Destination language

- Insert “native” code
- Use [protectspecialchars=off] to keep ODS from protecting text from interpretation by the program.
- Documentation
  - (SAS RTF) From http://support.sas.com, on the left side, click *Focus Areas*, *Base SAS, ODS, SAS notes for ODS*, and under the ODS RTF section, click *Concepts*
  - (Microsoft RTF) http://msdn2.microsoft.com/en-us/library/aa140277(office.10).aspx
  - (W3 HTML) www.w3schools.com

See documentation

Before

```
define treatment / across ‘Treatment Group’ order = internal format = treat. ;
```

After

```
define treatment / across ‘Treatment Group \brdrb\brdrs’
style=[protectspecialchars=off] order = internal format = treat. ;
```

Before

```
define text / group ‘ ‘style = [cellwidth=3 in just=left indent=.25 in]  ;
```

After

```
define text / group ‘ ‘style = [cellwidth=3 in just=left pretext='\ql\li360 '
protectspecialchars=off]  ;
```

Note: 1 inch = 1,440 twibs

See demo5rtf1.rtf

In-line Formatting

- Introduced in 8.2 as a way to style and format text “on the fly”
- Not PROC-specific
- Components
    - Escape character
    - Inline function
    - argument
- Documentation stinks!
    - Check out http://support.sas.com/rnd/base/topics/expv8/inline82.html
    - From http://support.sas.com, on the left side, click *Focus Areas*, *Base SAS, ODS*, then scroll to the bottom under the Archive section. Click *In-Line Formatting For ODS in SAS 8.2*

See documentation

Before

```
define treatment / across ‘Treatment Group \brdrb\brdrs’
style=[protectspecialchars=off] order = internal ;
```

After

```
ODS escapechar = “^” ; /* global statement – can be executed anywhere */
define treatment / across ‘Treatment Group ^R/RTF”\brdrb\brdrs”’ order = internal ;
```

Before

```
compute before order1 / style=[font_weight=bold];
        line @1 ‘ ‘;
        line @1 order1 head. ;
endcomp;
```

After

```
compute before order1 / style=[font_weight=bold];
        line @1 “^n” order1 head. ;
endcomp;
```

See demo6rtf1.rtf

Conditional Styling

- COMPUTE Blocks allow certain data step statements, including IF-THEN.
- CALL DEFINE sets attributes of report columns

```
compute text ;
        if order1 eq 3 and order2 in (4,5) then
          call define(_col_,'style','style=[pretext="\ql\li720" font_style=italic]');
endcomp;
```

```
compute text ; /* just for illustration */
        if order1 eq 3 and order2 eq 1 then
          call define(_col_,'style','style=[pretext="\ql\li720" font_style=italic]');
endcomp;
```

See documentation
See demo7cf1.rtf
See demo8cf2.rtf
See documentation

```
compute sales ;
   if department eq ‘p1' and sector eq 'nw' then
    call define (_col_,'style','style=[background=red  font_weight=bold]');
endcomp;
```

Sector	Dept	Total Sales
ne	np1	$290.00
	np2	$840.00
	p1	$490.00
	p2	$211.00
nw	np1	$150.00
	np2	$1,070.00
	p1	$1,055.00
	p2	$179.00

Sector	Dept	Total Sales
ne	np1	$290.00
	np2	$840.00
	p1	$490.00
	p2	$211.00
nw	np1	$150.00
	np2	$1,070.00
	p1	$1,055.00
	p2	$179.00

```
compute sales ;
   if department eq ‘np1' and sector eq 'nw' then
    call define (_col_,'style','style=[background=red  font_weight=bold]');
endcomp;
```

Sector	Dept	Total Sales
ne	np1	$290.00
	np2	$840.00
	p1	$490.00
	p2	$211.00
nw	np1	$150.00
	np2	$1,070.00
	p1	$1,055.00
	p2	$179.00

Sector	Dept	Total Sales
ne	np1	$290.00
	np2	$840.00
	p1	$490.00
	p2	$211.00
nw	np1	$150.00
	np2	$1,070.00
	p1	$1,055.00
	p2	$179.00

Temporary Variables

- **In the DATA step**
  - **Automatic variables (e.g. _ERROR_, _N_)**
  - **Initialized by you**
    - **Data set options (e.g. IN=, END=)**
    - **BY statement (FIRST., LAST.)**

- **IN PROC REPORT**
  - **Initialized by you in a COMPUTE block**
  - **Disappears after the procedure**
  - **Does not appear in a COLUMN or DEFINE statement**
  - **RETAINS ITS VALUE**

Sector	Dept	Total Sales
ne	np1	$290.00
	np2	$840.00
	p1	$490.00
	p2	$211.00
ne		$1831.00
nw	np1	$150.00
	np2	$1,070.00
	p1	$1,055.00
	p2	$179.00
nw		$2454.00

```
ods rtf file='example1.rtf' ;
proc report nowindows data=grocery ;
column sector department sales=salestotal sales=percent ;

define sector          /  'Sector' group;
define department  / ‘Dept’ group center;
define salestotal     / 'Total/Sales‘ sum format=dollar9.2 center ;
define percent         / 'Percent/of Total‘ pctsum
                         format=percent8.2 center;
break after sector / summarize ;
run;
ods rtf close ;
```

Sector	Dept	Total Sales	Percent of Total
ne	np1	$290.00	6.77%
	np2	$840.00	19.60%
	p1	$490.00	11.44%
	p2	$211.00	4.92%
ne		$1,831.00	42.73%
nw	np1	$150.00	3.50%
	np2	$1,070.00	24.97%
	p1	$1,055.00	24.62%
	p2	$179.00	4.18%
nw		$2,454.00	57.27%

Before

```
column sector dept sales=salestotal sales=percent ;
define percent   /  'Percent/of Total‘ pctsum
                          format=percent8.2 center;
```

Sector	Dept	Total Sales
ne	np1	$290.00
	np2	$840.00
	p1	$490.00
	p2	$211.00
ne		$1,831.00
nw	np1	$150.00
	np2	$1,070.00
	p1	$1,055.00
		$179.00
		$2,454.00

After

```
column sector dept sales=salestotal percent ;
define percent    /  'Percent /of Sector‘
computed format=percent8.2 center;
```

```
Compute before sector ;
            sectortotal = salestotal ;
endcomp ;
```

```
Compute percent;
   percent = salestotal / sectortotal ;
endcomp ;
```

Sector	Dept	Total Sales	Percent of Sector
ne	np1	$290.00	15.84%
	np2	$840.00	45.88%
	p1	$490.00	26.76%
	p2	$211.00	11.52%
ne		$1,831.00	100.0%
nw	np1	$150.00	6.11%
	np2	$1,070.00	43.60%
	p1	$1,055.00	42.99%
	p2	$179.00	7.29%
nw		$2,454.00	100.0%

```
proc report nowindows data=temp.demo

   style = [frame=hsides rules=groups posttext="Explanation of p values"
            font_face="times new roman" font_size=8 pt  just=left ]

   style(header lines) = [font_face="times new roman"]

   style(column) = [font_face="times new roman" cellwidth=1.25 in just=center]  ;

column order1 order2 text treatment,counts ('p value' pvalue) ;
define order1 / group noprint order=internal ;
define order2 / group noprint order=internal ;
define text / group ' 'style=[just=left cellwidth=3 in pretext='\ql\li360' protectspecialchars=off];
define treatment / across 'Treatment Group ^R/RTF"\brdrb\brdrs"'
  order = internal ;
define counts / group ' ' ;
define pvalue / sum ' ' format=pval.  ;

compute before order1 / style = [font_weight=bold]  ;
temp = order1 ;
line @1 '^n' order1 head. ;
endcomp;
```

See demo9cf3.rtf

```
compute text ;
if temp eq 3 and order2 in (4,5) then call define(_col_,'style','style=[pretext="\ql\li720"
font_style=italic]');
endcomp;
```

Web Sites

- (SAS RTF) From http://support.sas.com, on the left side, click *Focus Areas*, *Base SAS, ODS, SAS notes for ODS*, and under the ODS RTF section, click *Concepts*
- (Microsoft RTF) http://msdn2.microsoft.com/en-us/library/aa140277(office.10).aspx
- (W3 HTML) www.w3schools.com
- (SAS Online doc') From http://support.sas.com, click *Product Documentation* then click on the version you want
- (SAS In-Line Formatting) From http://support.sas.com, on the left side, click *Focus Areas*, *Base SAS, ODS*, then scroll to the bottom under the Archive section. Click *In-Line Formatting For ODS in SAS 8.2*

Contact Information

Name:	Mike Molter
Phone Number:	(919) 462-4199 (PPD)
E-Mail 1:	mike.molter@rtp.ppdi.com (PPD)
E-Mail 2:	mmolter001@nc.rr.com (home)
E-Mail 3:	molter.mike@gmail.com (home)

Thank You PPD!!!